

Knowledge Graphs

Lecture 3 – Querying Knowledge Graphs with SPARQL

3.4 SPARQL Sub-Select and Property Paths

Prof. Dr. Harald Sack
FIZ Karlsruhe – Leibniz Institute for Information Infrastructure
AIFB – Karlsruhe Institute of Technology
Autumn 2023

Karlsruher Institut für Technologie

FIZ Karlsruhe
Leibniz-Institut für Informationsinfrastruktur

Knowledge Graphs

Lecture 3: Querying Knowledge Graphs with SPARQL


CUUCH
MCARIS
what scifi books were written by the 200 most influential authors?

[1]

SPARQL Sub-Selects

Example: what scifi books were written by the 200 most influential authors?

```
PREFIX wd: <http://www.wikidata.org/entity/>
PREFIX wdt: <http://www.wikidata.org/prop/direct/>
PREFIX wikibase: <http://wikiba.se/ontology#>
PREFIX bd: <http://www.bigdata.com/rdf#>

SELECT ?influencerLabel ?bookLabel ?authorCount
WHERE {
  { SELECT ?influencer (COUNT(?author) AS ?authorCount)
    WHERE {
      ?author wdt:P106 wd:Q36180 ;     # ?author :occupation :Writer
              wdt:P737 ?influencer .   #         :influencedBy ?influencer
      ?influencer wdt:P106 wd:Q36180 .# ?influencer :occupation :Writer
    }  GROUP BY ?influencer ORDER BY DESC(?authorCount)
       LIMIT 200
  }
  ?influencer wdt:P800 ?book .        # ?influencer :notableWork ?book
  ?book       wdt:P136 wd:Q24925 .    # ?book :genre :ScienceFiction
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en" }
} ORDER BY DESC(?authorCount) ?influencerLabel
```

inner Subquery

- Subqueries are a way to embed SPARQL queries within other queries.
- Result is achieved by first evaluating the inner query.

WIKIDATA [2]

query SPARQL endpoint

SPARQL Sub-Selects

Example: **what scifi books** were written by the **200 most influential authors**?

Wikidata Query Service | Examples | Help | More tools | Query Builder | English

```
PREFIX wd: <http://www.wikidata.org/entity/>
PREFIX wdt: <http://www.wikidata.org/prop/direct/>
PREFIX wikibase: <http://wikiba.se/ontology#>
PREFIX bd: <http://www.bigdata.com/rdf#>

SELECT ?influencerLabel ?bookLabel ?authorCount
WHERE {
  { SELECT ?influencer (COUNT(?author) AS ?authorCount)
    WHERE {
      ?author wdt:P106 wd:Q36180 ;     # ?author :occupation :Writer
              wdt:P737 ?influencer .   #         :influencedBy ?influencer
      ?influencer wdt:P106 wd:Q36180 .# ?influencer :occupation :Writer
    } GROUP BY ?influencer ORDER BY DESC(?authorCount)
      LIMIT 200
  }
  ?influencer wdt:P800 ?book .       # ?influencer :notableWork ?book
  ?book wdt:P136 wd:Q24925 .   # ?book :genre :ScienceFiction
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en" }
} ORDER BY DESC(?authorCount) ?influencerLabel
```

Table | 49 results in 932 ms | Code | Download | Link

influencerLabel	bookLabel	authorCount
Robert Louis Stevenson	Strange Case of Dr Jekyll and Mr Hyde	20
H. P. Lovecraft	The Case of Charles Dexter Ward	17
H. P. Lovecraft	Beyond the Wall of Sleep	17
H. P. Lovecraft	At the Mountains of Madness	17
H. P. Lovecraft	The Shadow Over Innsmouth	17
H. P. Lovecraft	The Call of Cthulhu	17
Cormac McCarthy	The Road	11
H. G. Wells	The War of the Worlds	11
H. G. Wells	The Time Machine: An Invention	11
H. G. Wells	The Shape of Things to Come	11
H. G. Wells	The Island of Dr Moreau	11

WIKIDATA [2]

query SPARQL endpoint

SPARQL Sub-Selects

Example: what scifi books were written by the 200 most influential authors?

```
#defaultView:Graph
PREFIX wd: <http://www.wikidata.org/entity/>
PREFIX wdt: <http://www.wikidata.org/prop/direct/>
PREFIX wikibase: <http://wikiba.se/ontology#>
PREFIX bd: <http://www.bigdata.com/rdf#>

SELECT  ?influencer ?influencerLabel ?book ?bookLabel ?authorCount
WHERE {
  { SELECT ?influencer (COUNT(?author) AS ?authorCount)
    WHERE {
      ?author wdt:P106 wd:Q36180 ;    # ?author :occupation :Writer
              wdt:P737 ?influencer .  #         :influencedBy ?influencer
      ?influencer wdt:P106 wd:Q36180 .# ?influencer :occupation :Writer
    }  GROUP BY ?influencer ORDER BY DESC(?authorCount)
       LIMIT 200
  }
  ?influencer wdt:P800 ?book .       # ?influencer :notableWork ?book
  ?book       wdt:P136 wd:Q24925 .   # ?book :genre :ScienceFiction
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en" }
} ORDER BY DESC(?authorCount) ?influencerLabel
```

With the Wikidata SPARQL endpoint, we are able to display the result as a graph.

WIKIDATA [2]

query SPARQL endpoint

SPARQL Sub-Selects

Example: **what scifi books** were written by the **200 most influential authors**?



With the Wikidata SPARQL endpoint, we are able to display the result as a graph.

WIKIDATA [2]

query SPARQL endpoint

SPARQL Property Paths

A **property path** is a possible route through an RDF graph between two graph nodes.

- trivial case: property path of length 1, i.e. a triple pattern
- **alternatives**: match one or both possibilities

```
{ :book1 dc:title|rdfs:label ?displayString . }
```

- **sequence**: property path of length >1

```
{ :alice foaf:knows/foaf:knows/foaf:name ?name . }
```

- **inverse property paths**: reversing the direction of the triple

```
{ ?x foaf:mbox <mailto:alice@example> . }
                    =
{ <mailto:alice@example> ^foaf:mbox ?x . }
```

SPARQL Property Paths

inverse path sequences

```
{ ?x foaf:knows/^foaf:knows ?y .
    FILTER (?x != ?y) }
```



Example:
Who else besides me knows the people I know?

SPARQL Property Paths

arbitrary length match

```
{ :alice foaf:knows+/foaf:name ?name .      }
```



Example:
what are the names of all persons I know and those who they know (and so on...) ?

SPARQL Property Paths

- **inverse path sequences**

```
{ ?x foaf:knows/^foaf:knows ?y .
    FILTER (?x != ?y) }
```

- **arbitrary length match**

```
{ :alice foaf:knows+/foaf:name ?name .      }
```

- **negated property paths**

```
{ ?x !(rdf:type|^rdf:type) ?y . }
```

SPARQL Property Paths

Example: who else was influenced by the influencers of H. P. Lovecraft ?

```
PREFIX wd: <http://www.wikidata.org/entity/>
PREFIX wdt: <http://www.wikidata.org/prop/direct/>
PREFIX wikibase: <http://wikiba.se/ontology#>
PREFIX bd: <http://www.bigdata.com/rdf#>

SELECT  ?influencedByInfluencersLabel
WHERE {
  wd:Q169566 wdt:P737|^wdt:P737 ?influencedByInfluencers
  SERVICE wikibase:label { bd:serviceParam wikibase:language "en" }
}
```

WIKIDATA [2]

query SPARQL endpoint

SPARQL Property Paths

Example: who else was influenced by the influencers of H. P. Lovecraft ?



WIKIDATA [2]

query SPARQL endpoint

[3]

Next Lecture...

Bibliographic References:

- Steve Harris, Andy Seaborne (2013), *SPARQL 1.1 Query Language*, W3C Recommendation 21 March 2013
- Aidan Hogan (2020), *The Web of Data*, Springer.
  - Chap. 6.2.9 Sub-Select Queries, 365–367.
  - Chap. 6.2.6 Property Graphy, 351–355.

Picture References:

[1] "A science fiction movie poster for "Cthulhu and the Gods of Mars" which depicts the first landing of humans on Mars in a retro-futuristic style showing how the great Cthulhu is hovering over the red dessert facing a few human astronauts surrounded by strange ancient artefacts.", created via ArtBot, Deliberate, 2023, [CC-BY-4.0], https://tinybots.net/artbot

[2] Wikidata logo, Wikimedia Commons [Public Domain], https://commons.wikimedia.org/wiki/File:Wikidata-logo-en.svg

[3] "A movie poster for the science fiction novel "the first Men on the Mars" which depicts the first landing on Mars in a retro-futuristic style showing the red dessert like surface of Mars, the rocket landing ship, a few astronauts and a Martian rover.", created via ArtBot, Deliberate, 2023, [CC-BY-4.0], https://tinybots.net/artbot